Ejemplos de búsqueda incremental utilizando el programa busqueda_incremental.sce editado en Scilab
﻿
Archivo de comandos (script) busqueda_incremental.sce
Utiliza las functions func.sce e incsearch.sce

clear
close
// Universidad Tecnológica Nacional- Facultad Regional Rosario
// Departamento de Ingeniería Química
// Matemática Superior Aplicada
//
//Ingreso de fichero de datos con los valores de xmin, xmax y ns
load('datos.dat')
disp('Extremo inferior del intervalo de búsqueda, xmin: '),xmin
disp('superior del intervalo de búsqueda, xmax: '),xmax
disp('Número de subintervalos, ns: '),ns
//
function y=func(x)
    y=x-4*sin(x)
endfunction
//
function xb = incsearch(func,xmin,xmax,ns)
// incsearch: Localizador de raices por búsqueda incremental
//   xb = incsearch(func,xmin,xmax,ns):
//      Encuentra intervalos de variación de la variable independiente x 
//      cuyos extremos presentan cambios de signo de una función real 
//      definida sobre un cierto dominio
// Entradas:
//   func = Nombre de la función
//   xmin, xmax = Extremos del intervalo
//   ns = Número de subintervalos (default = 50)
// Salidas:
//   xb(k,1) Es el extremo inferior correspondiente al k-ésimo intervalo 
//   que presenta cambio de signo
//   xb(k,2) Es el extremo superior correspondiente al k-ésimo intervalo 
//   que presenta cambio de signo
//   Si no se encuentran intervalos, xb = [].
[lhs, rhs] = argn()
lhs = argn(1);
rhs = argn(2);
if rhs < 4, ns = 50; end //Si no se ingresa ns, se lo fija en el valor 50
// Búsqueda incremental
x = linspace(xmin,xmax,ns); 
f = func(x);
plot(x,f);
[bookmark: _GoBack]set(gca(),'grid',[1 1])
nb=0; xb=[]; //xb es nulo a menos que se detecte un cambio de signo
for k = 1:length(x)-1
  if sign(f(k))~= sign(f(k+1)) //Verifica si se produce un cambio de signo
    nb=nb+1;
    xb(nb,1)=x(k);
    xb(nb,2)=x(k+1);
  end
end
if isempty(xb)    //Informa que no se encontraron intervalos donde la función cambia de signo
  disp('No se encontraron intervalos')
  disp('Verificar el intervalo de búsqueda o incrementar ns')
else
  disp('Número de subintervalos:') //Muestra el número de subintervalos
  disp(nb)
end
endfunction
xb=incsearch(func,xmin,xmax,ns)

En primer lugar se carga en el fichero datos.dat los valores del intervalo de búsqueda, xmin y xmax y el nro. de subintervalos, ns, que requiere el programa incsearch.  
Para  la función f(x) = x – 4sen(x), se utilizarán los siguientes datos:
--> xmin = -2*%pi
--> xmax = 2*%pi
--> ns = 100.
 --> save('datos.dat', 'xmin', 'xmax', 'ns')
Luego, se ejecuta (con echo) el programa busqueda_incremental.sce para la función cuyas raíces se desea localizar. Si se desea efectuar la localización de las raíces de otra función se deberán redefinir los datos del fichero de datos y editar la nueva función.
Extremo inferior del intervalo de búsqueda, xmin: 
 xmin  =  -6.2831853
 Extremo superior del intervalo de búsqueda, xmax: 
 xmax  = 6.2831853
 Número de subintervalos, ns: 
 ns  =  100.

Resultado
Número de subintervalos: 3
xb  = 
  	-2.4751942  -2.3482612
  	-0.0634665   0.0634665
   	 2.3482612   2.4751942


Se ejecuta el programa para diferentes polinomios
1) P1(x) = x^4+6*x^3+7*x^2-6*x-8
Extremo inferior del intervalo de búsqueda, xmin: 
xmin  =  -5.
Extremo superior del intervalo de búsqueda, xmax: 
xmax  =   2.
Número de subintervalos, ns: 
ns  =  50.

Resultado
 	Número de subintervalos: 4
xb  = 
  	-4.1253237  -3.9983907
  	-2.0943951  -1.9674621
  	-1.0789308  -0.9519978
   	 0.9519978   1.0789308

Raíces del polinomio: 
x1 = 1
x2 = -1
x3 = -2
x4 = -4



2) P2(x) = x^4+7*x^3+12*x^2-4*x-16

Extremo inferior del intervalo de búsqueda, xmin: 
xmin  =  -5.

Extremo superior del intervalo de búsqueda, xmax: 
xmax  = 2.

Número de subintervalos, ns: 
ns  =  100.

Resultado
Número de subintervalos: 4
xb  = 
  -4.010101    -3.9393939
   0.9393939    1.0101010

Raíces del polinomio: 
x1 = 1
x2,3 = -2 (raíz doble)
x4 = -4




3)	P3(x) = x^4-6*x^3+18*x^2-30*x+25

Extremo inferior del intervalo de búsqueda, xmin: 
xmin  =  -2.

Extremo superior del intervalo de búsqueda, xmax: 
xmax  = 2.

Número de subintervalos, ns: 
ns  =  50.

Resultado
No se encontraron intervalos

Verificar el intervalo de búsqueda o incrementar ns
xb  = []

Raíces del polinomio: No tiene raíces reales
X1,2 = 1 ± 2i
X3,4 = 2 ± i





4) P4(x) = x^4+x^3-5*x^2+23*x-20

Extremo inferior del intervalo de búsqueda, xmin: 
xmin  =  -5.

Extremo superior del intervalo de búsqueda, xmax: 
xmax  = 2.

Número de subintervalos, ns: 
ns  =  50.

Resultado
Número de subintervalos: 100.
 xb  = 
  	-4.0101010   -3.9393939
   	 0.9393939    1.0101010

Raíces del polinomio: 
x1 = 1
x2,3 = 1 ± 2i
x4 = -4

